

# SCIENCE & VIE

MENSUEL

N° 849 JUIN 1988

**Greffes de foie, rein,  
cœur, poumon :  
maintenant la routine**

**Tchernobyl aujourd'hui :  
les analyses  
de Science & Vie**

**Auto :  
le super  
obligatoire !**

**GUERRE DES ÉTOILES:  
1<sup>re</sup> MANŒUVRE  
RÉUSSIE !**

M 2578 - 849 - 17,00 F



3792578017005 08490

120 FB - 5 FS - Scan 3.50 - 500 Ptas - 18 Dh - 2.300 Dt - 4000 L - USA NYC \$3.75 - OTHER \$ 3.95

# Un jeu vidéo compatible "Captain Power"

## ÉLECTRONIQUE AMUSANTE

La série télévisée "Captain Power" actuellement proposée par La Cinq, inclut un codage permettant de l'utiliser en tant que jeu vidéo. Des jouets, capables d'analyser ce codage, sont commercialisés par Mattel. Cependant, leur principe de fonctionnement reste relativement simple. Nous vous proposerons donc de réaliser un petit appareil, auquel vous pourrez donner la forme que vous souhaitez, sous réserve de respecter certains impératifs, pouvant être utilisé avec ce type de codage.

Notons que nous vous proposons également, dans notre rubrique Informatique amusante, un programme permettant d'une part de contrôler le bon fonctionnement de ce montage et, d'autre part, de l'utiliser en tant que jeu vidéo sans devoir passer par l'intermédiaire de la série télévisée. Cette réalisation comportant cependant un nombre relativement important de composants, nous vous la proposerons en deux parties. La première concernera la détection des signaux lumineux émis par le téléviseur et la seconde leur conversion en nombre de points.

Voyons donc le principe optique utilisé et la conversion du codage

en signaux électriques dans ses grandes lignes, puis comment le réaliser.

Ici notre optique sera on ne peut plus simple. En effet, en tant qu'objectif, nous n'utiliserons qu'un simple carton noir percé d'un trou de 3 mm environ. Notre capteur sera une photorésistance — LDR — placée à 15 cm au moins de notre objectif. Il va de soi que le tube reliant notre cellule à son diaphragme devra être étanche à la lumière, donc de préférence noir ou de couleur sombre. Notons à ce propos que les LDR sont peu sensibles au rouge. Il pourra donc éventuellement être rouge.

Quoi qu'il en soit, nous devons être en mesure de transformer le scintillement de l'image présentée en signal électrique. Pour cela, notre cellule LDR sera câblée en série avec une résistance de forte valeur. Un condensateur nous permettra de prélever la composante alternative du signal électrique ainsi obtenu. Ce nouveau signal permettra de piloter le reste de ce montage.


Rappelons en effet que la visée d'une image vidéo classique fournira sur notre cellule un signal à une fréquence de 50 Hz, une cible

### NOMENCLATURE

- $R_1 = 4,7 \text{ K}\Omega$  (jaune, violet, rouge, or)
- $R_2 = 47 \text{ K}\Omega$  (jaune, violet, orange, or)
- $R_3 = 4,7 \text{ K}\Omega$  (jaune, violet, rouge, or)
- $R_4 = 4,7 \text{ K}\Omega$  (jaune, violet, rouge, or)
- $R_5 = 2,2 \text{ K}\Omega$  (rouge, rouge, rouge, or)
- $R_6 = 470 \text{ K}\Omega$  (jaune, violet, jaune, or)
- $R_7 = 470 \text{ K}\Omega$  (jaune, violet, jaune, or)
- $R_8 = 22 \text{ K}\Omega$  (rouge, rouge, orange, or)
- $R_9 = 47 \text{ K}\Omega$  (jaune, violet, orange, or)
- $R_{10} = 4,7 \text{ K}\Omega$  (jaune, violet, rouge, or)
- $R_{11} = 2,2 \text{ K}\Omega$  (rouge, rouge, rouge, or)
- $R_{12} = 2,2 \text{ K}\Omega$  (rouge, rouge, rouge, or)
- $R_{13} = 2,2 \text{ K}\Omega$  (rouge, rouge, rouge, or)
- $R_{14} = 47 \text{ K}\Omega$  (jaune, violet, orange, or)
- $R_{15} = 4,7 \text{ K}\Omega$  (jaune, violet, rouge, or)
- $R_{16} = 47 \text{ K}\Omega$  (jaune, violet, orange, or)
- $C_1 = 22 \text{ microfarads } 12 \text{ volts}$
- $C_2 = 10 \text{ nanofarads}$
- $C_3 = 22 \text{ microfarads } 12 \text{ volts}$
- $C_4 = 330 \text{ picofarads}$
- $C_5 = 22 \text{ nanofarads}$
- $C_6 = 4,7 \text{ microfarads } 12 \text{ volts}$
- $C_7 = 22 \text{ microfarads } 12 \text{ volts}$
- $C_8 = 330 \text{ picofarads}$
- $C_9 = 4,7 \text{ microfarads } 12 \text{ volts}$
- $IC_1 = \text{LM } 324 \text{ ou équivalent}$
- LDR = cellule photo-résistance
- $D_1 = 1\text{N } 4148$
- $D_2 = 1\text{N } 4148$
- Plaque de câblage ; fil blindé

"gagnante", dans le cas de notre codage "Captain Power", un signal à 25 Hz, et un tir ennemi un signal de 12,5 Hz. Certes la cellule LDR n'est pas capable de fournir directement des signaux électriques compatibles avec les exigences de notre montage. Il faudra donc commencer par les amplifier.


### IMPLANTATION DES COMPOSANTS



Il ne sera pas étonnant de trouver notre cellule immédiatement suivie d'un amplificateur opérationnel. La fonction première de celui-ci sera l'amplification de la composante alternative du signal électrique délivré par la cellule. Notons que son gain en tension, c'est-à-dire son facteur d'amplification a été prévu pour des conditions normales d'éclairément. Il ne faudra donc pas être surpris si notre montage ne fonctionne pas lorsque le téléviseur est placé en plein soleil et que l'image reste à peine perceptible sur l'écran. N'oublions pas ici que le codage reste purement optique ! Sur la sortie de notre amplificateur nous obtiendrons un signal alternatif, dont la fréquence sera soit de 50 Hz, si la portion d'écran visée ne comporte pas d'information spécifique, soit de 25 Hz si elle correspond à une cible, soit de 12,5 Hz s'il s'agit d'une zone dangereuse. Un oscillateur, dont la fréquence sera calibrée et servira de référence, associée à un compteur d'impulsions lumineuses, permettra de différencier chaque cas.

Tel sera le principe que nous utiliserons pour présenter au compteur de points, que nous vous proposerons de réaliser le mois prochain, les informations "actives" contenues dans l'image de votre téléviseur.

La réalisation de cette première partie du montage ne doit pas poser de problème particulier, sauf en ce qui concerne la partie optique. Seul impératif : réaliser un



#### OÙ SE PROCURER LES COMPOSANTS

△ MAGNETIC FRANCE, 11 place de la Nation, 75011 Paris, tél. (1) 43 79 39 88

△ PENTASONIC, 10 boulevard Arago, 75013 Paris, tél. 43 36 26 05


△ T.S.M., 15 rue des Onze-Arpents, 95130 Franconville, tél. 34 13 37 52

△ URS MEYER ELECTRONIC, 2 052 Fontainemelon Suisse.

△ Ces composants sont également disponibles chez la plupart des revendeurs régionaux.

tube noir, de préférence de 15 cm de long, dont l'une des extrémités portera en son centre la LDR et l'autre un trou de 3 mm environ. Ceci constitue la partie essentielle de notre réalisation ! Le reste du câblage restera des plus simples. Il faudra cependant prendre soin de respecter la polarité des condensateurs, ainsi que le brochage des circuits intégrés. De même nous vous rappelons que l'ensemble des bandes cuivrées de la plaquette de câblage, devra être coupé sous chaque circuit intégré.

Nous vous donnons rendez-vous le mois prochain pour terminer ce jeu vidéo. **Henri-Pierre Penel**



# Alerte aux OVNI

## INFORMATIQUE AMUSANTE

Nous vous proposons ce mois-ci un petit programme de jeu vidéo. En fait, pour pouvoir l'utiliser il vous faudra disposer de deux éléments : d'un ordinateur Amstrad, équipé d'un moniteur couleur et, soit du jouet proposé par Mattel et destiné au feuilleton télévisé *Captain Power*, soit du montage décrit dans notre rubrique *Electronique amusante*. Ceci dit, voyons quelles seront les règles de ce jeu.

L'ordinateur déplace sur l'écran des soucoupes volantes. Lorsque



```

10 MODE 0
12 REM *****
13 REM *
14 REM * DETERMINATION DES COULEURS DU FOND. *
15 REM *
16 REM *****
20 PAPER 5: BORDER 0:CLS
23 REM *****
24 REM *
25 REM * CARACTERES GRAPHIQUES DES SOUCOUPES *
26 REM *
27 REM *****
30 LET A$=CHR$(214)+CHR$(143)+CHR$(143)+CHR$(215)
40 LET B$=CHR$(213)+CHR$(143)+CHR$(143)+CHR$(212)
50 LET C$=CHR$(214)+CHR$(215)
53 REM *****
54 REM *
55 REM * COULEURS DES TACHES SCINTILLANTES *
56 REM *
57 REM *****
60 INK 14,26,0
70 INK 15,6,24
73 REM *****
74 REM *
75 REM * CHOIX DU NIVEAU *
76 REM *
77 REM *****
80 PEN 1:LOCATE 1,5:PRINT "DE 0 A 10; 10=FACILE"
90 LOCATE 1,10:INPUT "VOTRE NIVEAU ";NI
100 IF NI<0 OR NI>10 THEN GOTO 90
103 REM *****
104 REM *
105 REM * CHOIX DE L'OPTION *
106 REM *
107 REM *****
110 LOCATE 1,15:INPUT "OPTION (O OU 1) ";OP
120 IF OP<0 OR OP>1 THEN GOTO 110
123 REM *****
124 REM *
125 REM * DEBUT DU JEU *
126 REM *
127 REM *****
130 CLS
140 PEN 3:LOCATE 9,1:PRINT C$
150 LOCATE 8,2:PRINT A$
160 LOCATE 8,3:PRINT B$
170 LET KC=10
173 REM *****
174 REM *
175 REM * INITIALISATION DES VARIABLES *
176 REM *
177 REM *****
180 LET YC=12
190 LET XF=10
200 LET YP=12
210 LET SA=0
220 LET SB=0
223 REM *****
224 REM *
225 REM * DEFFINITION DES SYMBOLES GRAPHIQUES *
226 REM *
227 REM *****
230 SYMBOL AFTER 200
240 SYMBOL 214,7,31,63,127,127,255,255,255
250 SYMBOL 215,224,248,252,254,254,255,255,255
260 SYMBOL 213,255,255,255,127,127,63,31,7
270 SYMBOL 212,255,255,255,254,254,252,248,224
273 REM *****
274 REM *
275 REM * MISE EN PLACE DES FENETRES *
276 REM *
277 REM *****
280 WINDOW #1,9,10,4,5
290 WINDOW #2,8,12,5,8
300 WINDOW #3,5,15,8,12
310 WINDOW #4,2,17,12,17
320 WINDOW #5,3,19,17,23
330 WINDOW #6,1,20,23,25
333 REM *****
334 REM *
335 REM * DEFFINITION DES ENVELOPPES SONORES *
336 REM *
337 REM *****
340 ENT -1,15,30,1,15,-30,1,15,-10,1
350 ENT -2,1,-15,1,15,10,1,1,0,1
360 ENV 1,1,15,1,1,0,5,15,-1,10
370 ENV 2,1,15,1,1,0,5,15,-1,2
373 REM *****
374 REM *
375 REM * TIRAGE ALEATOIRE DE L'ETAT DES CIBLES *
376 REM *
377 REM *****
380 LET CB=INT (RND*(NI+3))
390 SPEED INK 1,1
400 IF CB=2 AND SA+SB+OP>1 THEN PEN 14:GOSUB 770:L
ET SA=0:LET SB=0
410 IF CB=0 THEN LET SA=1
420 IF CB=1 THEN LET SB=1
430 GOSUB 470
440 IF SA+SB<>0 THEN FOR T=1 TO 30*NI:NEXT T

```

celles-ci deviennent blanches, il faut les viser et, si possible, les atteindre. Cependant la soucoupe-mère veille sur elles, et lorsqu'elles deviennent vulnérables, elle tente de les couvrir. Les tirs qu'elle émet sont capables de vous atteindre et donc de vous faire perdre des points. Plusieurs niveaux de difficultés de jeu ainsi qu'une option sont proposés. Ces paramètres agissent essentiellement sur la vitesse de déplacement des soucoupes et sur le temps de réaction du vaisseau-mère.

Ces quelques règles fixées, passons donc à l'écriture du programme. Nous commencerons par demander à l'ordinateur de travailler en mode 0 ; ceci pour pouvoir profiter de l'ensemble de sa palette de couleurs. Le fond de l'écran sera noir, instructions de la ligne 20, et nous déterminerons 3 variables, A\$,

B\$ et C\$, que nous utiliserons ultérieurement pour l'affichage des soucoupes. Rappelons à présent que le codage utilisé par le jouet Mattel pour compter ou décompter ses points, est basé sur un changement de la fréquence de scintillement de l'écran. Nous créerons donc deux couleurs scintillantes aux lignes 60 et 70. La fréquence de scintillement de INK 14 sera de 25 Hz et déterminera donc les cibles gagnantes, celle de INK 15, de 12,5 Hz, sera utilisée pour les tirs destructeurs de la soucoupe-mère. Vient ensuite le choix du niveau et de l'option de jeu ; lignes 80 à 120.

Le jeu peut maintenant commencer. La soucoupe-mère sera affichée sur l'écran, lignes 140 à 160. L'ensemble des variables sera initialisé et les symboles graphiques redéfinis. De même nous mettrons en place 6 fenêtres que nous utili-

serons pour simuler les tirs ennemis et quatre enveloppes sonores seront déterminées afin de piloter les illustrations sonores. La variable CB, tirée aléatoirement par la ligne 380, déterminera l'état des cibles, vulnérables ou non, et commandera le tir de la soucoupe-mère. Notons que ce tirage prend en compte la variable NI, la valeur de cette dernière étant fonction du niveau choisi. Ensuite, des tests effectués autour de VB aiguilleront le programme vers diverses sous-routines. Notons que les variables SA et SB permettent de commander la vulnérabilité des soucoupes.

Les sous-routines sont au nombre de deux. La première, placée des lignes 470 à 750, assure l'animation des soucoupes. Deux variables, XM et YM détermineront la direction du déplacement à chaque instant du jeu, XC et YC leur posi-

```

450 GOTO 380
463 REM *****
464 REM *
465 REM * SOUS-ROUTINE D'ANIMATION SOUCOUPES *
466 REM *
467 REM *****
470 LET XH=INT(RND*3)-1
480 LET YH=INT(RND*3)-1
490 IF XH=0 OR YH=0 THEN GOTO 470
500 LET XC=XC+XH
510 IF XC<1 THEN LET XC=1:GOTO 470
520 IF XC>16 THEN LET XC=16:GOTO 470
530 LET DR=INT(RND*5)
540 IF DR=0 THEN GOTO 470
550 LET YC=YC+YH
560 IF YC<4 THEN LET YC=4:GOTO 470
570 IF YC>21 THEN LET YC=21:GOTO 470
580 LOCATE XP,YP:PRINT " "
590 LOCATE 17-XP,25-YP:PRINT " "
600 LOCATE XP,YP+1:PRINT " "
610 LOCATE 17-XP,25-YP+1:PRINT " "
620 LOCATE 17-XP,25-YP+2:PRINT " "
630 LOCATE XP,YP+2:PRINT " "
633.REM *****
634 REM *
635 REM * MISE EN POSITION CIBLE DE LA PREMIERE *
636 REM * SOUCOUCPE ET ILLUSTRATION SONORE *
637 REM *
638 REM *****
640 IF SA=1 THEN PEN 14:SOUND 1,405,30+(5*NI),7,0,
1,0
650 LOCATE XC+1,YC:PRINT C$
660 LOCATE XC,YC+1:PRINT A$
670 LOCATE XC,YC+2:PRINT B$
680 PEN 12
683 REM *****
684 REM *
685 REM * MISE EN POSITION CIBLE DE LA SECONDE *
686 REM * SOUCOUCPE ET ILLUSTRATION SONORE *
687 REM *
688 REM *****
690 IF SB=1 THEN PEN 14:SOUND 2,400,35+(5*NI),7,0,
1,0
700 LOCATE 18-XC,25-YP:PRINT C$
710 LOCATE 17-XC,25-YP+2:PRINT B$
720 LOCATE 17-XC,25-YP+1:PRINT A$
730 PEN 7
740 LET XP=XC:LET YP=YC
750 RETURN

763 REM *****
764 REM *
765 REM * SOUS ROUTINE D'ANIMATION DU TIR *
766 REM * DE LA BASE ENNEMIE. *
767 REM *
768 REM *****
770 SPEED INK 2,2
773 REM *****
774 REM *
775 REM * ILLUSTRATION SONORE *
776 REM *
777 REM *****
780 SOUND 2,40,50,7,0,2,0
790 SOUND 1,30,50,7,0,2,0
793 REM *****
794 REM *
795 REM * COMMANDE DE SCINTILLEMENT *
796 REM * DES FENETRES *

797 REM *
798 REM *****
800 FOR I=1 TO 6
810 PAPER #I,15
820 CLS #I
830 FOR t=1 TO 100:NEXT T
840 NEXT I

843 REM *****
844 REM *
845 REM * ILLUSTRATION SONORE *
846 REM *
847 REM *****
850 SOUND 1,400,0,0,1,0,15
860 SOUND 2,405,0,0,1,0,17
870 BORDER 24

873 REM *****
874 REM *
875 REM * RETOUR A LA COULEUR DE BASE *
876 REM * DES FENETRES *
877 REM *
878 REM *****
880 FOR I=1 TO 6
890 PAPER #I,5
900 CLS #I
910 BORDER 24-I
920 FOR t=1 TO 50:NEXT T
930 NEXT I
940 BORDER 0
950 RETURN

```

tion instantanée. Notons que deux variables auxiliaires XP et YP mémorisent la position précédente des soucoupes avant chaque nouvel affichage. Nous les utiliserons pour effacer l'ancienne position de chaque soucoupe après son déplacement. Cette dernière fonction est assurée par les lignes 580 à 630.

Vient l'affichage des soucoupes : les variables SA et SB seront contrôlées et, en fonction de leur valeur, la couleur des soucoupes sera fixée. De même une illustration sonore, indiquant leur vulnérabilité, sera commandée ou non. Cette sous-routine se termine par l'actualisation des variables XP et YP.

La seconde sous-routine assure les tirs du vaisseau-mère. Elle est placée entre les lignes 770 et 950. Nous commencerons par modifier la fréquence de scintillement de l'écran, de manière à ce qu'il soit considéré comme "destructeur" à la ligne 770. Une illustration sonore sera demandée puis nous réaliserons l'animation lumineuse du tir des lignes 800 à 840 en modifiant successivement la couleur des fenêtres définies en début de programme. Une nouvelle illustration sonore, imitant une explosion, sera demandée puis, des lignes 880 à 930, nous ramènerons la couleur de nos fenêtres à la normale, c'est-à-dire le noir. Durant cette dernière phase, le tour de l'écran scintillera ; afin de renforcer l'effet d'explosion.

L'utilisation de ce programme est des plus simples. Après l'avoir entièrement tapé puis envoyé RUN, votre choix concernant le niveau et l'option sera demandé. Dans les deux cas, votre réponse devra être validée en frappant ENTER. Précisons que le niveau agit essentiellement sur la vitesse de déplacement des soucoupes et l'option sur le temps de réaction du vaisseau-mère. Ce choix terminé, l'animation commencera. Etant donné qu'il n'y a aucune interaction du jouet sur le déroulement du programme, nous n'avons pas prévu de pouvoir l'interrompre directement sur frappe d'une touche, par exemple. Pour modifier le niveau ou l'option, il faudra donc le relancer. Pour cela, on l'interrompra en tapant deux fois sur la touche rouge ESC puis RUN sera de nouveau demandé.

**Henri-Pierre Penel**